# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python2-xattr
_pkgname=xattr
pkgver=0.9.7
pkgrel=2
pkgdesc='Python wrapper for extended filesystem attributes.'
url='https://pypi.org/project/xattr/#files'
arch=('x86_64' 'aarch64')
license=('MIT')
depends=('python2' 'python2-cffi')
makedepends=('python2-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('98473ad4ac270b747416e448f426eb55888ed1a641027f27fc9ced58eff29c6866de6b1d914628e9be1785a93dc857977d172e6bcbe5233e6449447dd08d9bd0')

build() {
  cd "$_pkgname-$pkgver"

  python2 setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python2 setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build

  mv "$pkgdir/usr/bin/$_pkgname" "$pkgdir/usr/bin/${_pkgname}2"
}

